
Abstract. Our main goal is to improve the dissertation of Kiril Popov, who
is a postgraduate of Henri Poincare. In this statue we give an application for
computing the perturbation function in the restricted three-body problem Sun,
Jupiter and Hecuba. For this we use the system for the manipulation of sym-
bolic and numerical expressions “Maxima”. We obtain the perturbation func-
tion using Fourier expansions of some functions of the radial distance and the
true anomaly in terms of the mean anomaly. The expansion of true anomaly
is done using Cauchy numbers. For this we give a new method to calculate
Cauchy numbers. This method leads to a variation of binomial transform and a
generalization of Pascal triangle. To compute the terms in the Taylor series we
use a new formula for the partial derivatives of a homogeneous function of two
variables.

1 Introduction

The statue presents an application for computing the perturbation function in
the restricted three-body problem Sun, Jupiter and Hecuba. It is necessary be-
cause we wish to improve well-known theory for the movement of the asteroid
108 Hecuba, developed by Kiril Popov as postgraduate of Henri Poincare [1]. It
is used the system for the manipulation of symbolic and numerical expressions
“Maxima” [2]. The main plane is the Jupiter’s orbit. Following [3], at first the
perturbation function is expanded in Macloren series about the Hecuba’s appli-
cate. After that each term is expanded in Taylor series about the semimajor axes
of Hecuba and Jupiter and the difference between theirs mean longitudes. The
derivatives in the Taylor series are expanded in Fourier series using Laplace’s
coefficients. For this it is used the formula for the partial derivatives of the
homogeneous function of two variables [4]. The true anomaly is expanded in
trigonometric series of the mean anomaly using Cauchy numbers [5]. Other
expansions are made using Bessel functions [3, 5, 6].

294



On the Movement of the Asteroid 108 Hecuba

2 Main Principles and Symbols in the Application

1. Basic formula for the perturbation function:

R

a1
=

1
Δ

− �r.�r1
r31

, (1)

where a1 is the semimajor axis of Jupiter, �r and �r1 are the radius-distances
of Hecuba and Jupiter, Δ is the distance between Hecuba and Jupiter.

2. Multiletter symbols are used for the quantities in the application.

3. e and el are the eccentricities of Hecuba and Jupiter.

4. Instead of inclination i between two orbits the quantity g, what is equal to
sin2 i/2, is used .

5. p is the order of e, el and g.

6. Upper limit in the Fourier series is q=2p, because the Jupiter’s mean mo-
tion is two times bigger then the Hecuba’s one.

7. M and Ml are the mean anomalies of Hecuba and Jupiter.

8. f and fl are the true anomalies of Hecuba and Jupiter.

9. w and wl are the arguments of pericenter of Hecuba and Jupiter.

10. Q and Ql are the longitudes of the ascending nodes of Hecuba and Jupiter.

11. The quantities u, ul and v are defined from the equations:

u = ro− 1, (2)

ul = rl − 1, (3)

v = (λ− λ1) − (l − l1), (4)

where ro (ro in the application) is the ratio of the projection of the radius-
distance on the main plane and the semimajor axis of Hecuba, rl (rl in
the application) is the ratio of the radius-distance and the semimajor axis
of Jupiter, λ and λ1 are the longitudes of Hecuba and Jupiter, l and l1 are
the mean longitudes of Hecuba and Jupiter.

12. af is the ratio of the semimajor axes of Hecuba and Jupiter.

13. zq is the square of the ratio of the applicate and the semimajor axis of
Hecuba.

14. Each time the unnecessary terms are separated.

15. The important and the final results are saved.

3 Application for Computing the Perturbation Function in the Re-
stricted Three-Body Problem Sun, Jupiter And Hecuba

1. Determination the order of accuracy.

p:4;

2. Computing Bessel functions for Hecuba.

for n:0 thru p+1 do
j[n]:sum((-1)ˆk*(n*e/2)ˆ(2*k+n)/(k!*(n+k)!),
k,0,?truncate(float(p-n+1)/2));

3. Computing Bessel functions for Jupiter.

for n:0 thru p+1 do
jl[n]:sum((-1)ˆk*(n*el/2)ˆ(2*k+n)/(k!*(n+k)!),
k,0,?truncate(float(p-n+1)/2));

4. Computing the derivatives of Bessel functions for Hecuba.

for n:1 thru p+1 do dj[n]:diff(j[n],e);

5. Computing the derivatives of Bessel functions for Jupiter.

for n:1 thru p+1 do djl[n]:diff(jl[n],el);

6. Calculating Cauchy numbers.

for m:0 thru p do for k:0 thru p-m do
(N[-m-k-2,m,k]:0,N[m+k+2,m,k]:0);
for k:0 thru p do for n:0 thru k do
N[2*n-k,0,k]:(-1)ˆ(k-n)*binomial(k,n);
for m:0 thru p-1 do for k:0 thru p-m-1 do
for n:0 thru m+k+1 do N[2*n-m-k-1,m+1,k]:
N[2*n-m-k-2,m,k]+N[2*n-m-k,m,k];

7. Expansion of the Hecuba’s true anomaly.

ff:sum(sum(sum(2*(2*n-m-k)ˆ(k-1)*(e/2)ˆ(m+k)*
N[2*n-m-k,m,k]*sin((2*n-m-k)*M)/k!,n,
?truncate(float((m+k)/2)+1),m+k),k,0,p-m),m,0,p);
f:M+expand(taylor(ff*(1-eˆ2)ˆ(1/2),e,0,p));

8. Expansion of the Jupiter’s true anomaly.

ffl:sum(sum(sum(2*(2*n-m-k)ˆ(k-1)*(el/2)ˆ(m+k)*
N[2*n-m-k,m,k]*sin((2*n-m-k)*Ml)/k!,n,
?truncate(float((m+k)/2)+1),m+k),k,0,p-m),m,0,p);
fl:Ml+expand(taylor(ffl*(1-elˆ2)ˆ(1/2),el,0,p));

9. Expansion of the ratio of the radius-distance and the semimajor axis of
Hecuba and the square of it.

r:sum(coeff(expand(1+1/2*eˆ2-sum(2*e/nˆ2

*dj[n]*cos(n*M),n,1,p+1)),e,k)*eˆk,k,0,p);
rq:sum(coeff(expand(rˆ2),e,k)*eˆk,k,0,p-2);

10. Expansion of the ratio of the radius-distance and the semimajor axis of
Jupiter.

rl:sum(coeff(expand(1+1/2*elˆ2-sum(2*el/nˆ2

*djl[n]*cos(n*Ml),n,1,p+1)),el,k)*elˆk,k,0,p);

11. Expansion of the sine of the Hecuba’s true anomaly.

sf[0]:1;
sf[1]:expand(taylor(2*sqrt(1-eˆ2)*
sum(dj[n]*sin(n*M)/n,n,1,p+1),e,0,p-2));
for n:2 thru p do
sf[n]:expand(trigreduce(expand(sum(coeff(expand(
sf[n-1]*sf[1]),e,k)*eˆk,k,0,p-2))));

12. Expansion of the cosine of the Hecuba’s true anomaly.

cf[0]:1;
cf[1]:sum(coeff(expand(-e+2*(1/e-e)*
sum(j[n]*cos(n*M),n,1,p+1)),e,k)*eˆk,k,0,p-2);
for n:2 thru p do
cf[n]:expand(trigreduce(expand(sum(coeff(
expand(cf[n-1]*cf[1]),e,k)*eˆk,k,0,p-2))));

13. Expansion of the square of the sine of the Hecuba’s latitude.

sb:expand(trigreduce(expand(
sf[1]*cos(w)+cf[1]*sin(w))));
sbq:expand(trigreduce(expand(sum(coeff(expand(
sbˆ2),e,k)*eˆk,k,0,p-2))));

st[0]:1;
st[1]:sum(sum(coeff(coeff(expand(
gˆ2*(1-gˆ2/4)*sbq),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p);
for n:2 thru p/2 do
st[n]:expand(trigreduce(expand(sum(sum(
coeff(coeff(expand(st[n-1]*st[1]),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p))));

14. Expansion of the cosine of the Hecuba’s latitude.

expand(taylor(sqrt(1-xˆ2),x,0,p));
expand(sum(coeff(%,x,2*k)*st[k],k,0,p/2));
ct:sum(sum(coeff(coeff(%,e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p);

15. Expansion of the square of the ratio of the applicate and the semimajor
axis of Hecuba.

zq[0]:1;
zq[1]:sum(sum(coeff(coeff(expand(trigreduce(
expand(rq*st[1]))),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p);
for n:2 thru p/2 do
zq[n]:expand(trigreduce(expand(sum(sum(
coeff(coeff(expand(zq[n-1]*zq[1]),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p))));
save(zq,zq);

16. Expansion of the ratio of the projection of the radius-distance on the main
plane and the semimajor axis of Hecuba.

ro:expand(trigreduce(expand(sum(sum(
coeff(coeff(expand(r*ct),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p))));

17. Computing the quantities u and ul and theirs multiplications.

u[0]:1;
u[1]:expand(ro-1);
for n:2 thru p do
u[n]:expand(trigreduce(expand(sum(sum(
coeff(coeff(expand(u[n-1]*u[1]),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p))));

ul[0]:1;
ul[1]:expand(rl-1);
for n:2 thru p do
ul[n]:expand(trigreduce(expand(sum(
coeff(expand(ul[n-1]*ul[1]),el,k)

*elˆk,k,0,p))));
for k:0 thru p do for l:0 thru (p-k) do
us2[k,l]:expand(u[k]*ul[l]);
for k:0 thru p do for l:0 thru (p-k) do
us1[k,l]:sum(sum(sum(coeff(coeff(coeff(
us2[k,l],g,x),e,y),el,z)

*gˆx*eˆy*elˆz,x,0,p-z-y),y,0,p-z),z,0,p);
for k:0 thru p do for l:0 thru (p-k) do
us[k,l]:expand(trigreduce(us1[k,l]));

18. Expansion of the Hecuba’s longitude.

F:fv+w;
F+(log(1-gˆ2/4+gˆ2/4*%Eˆ(-2*%I*F))-
log(1-gˆ2/4+gˆ2/4*%Eˆ(2*%I*F)))/(2*%I);
expand(taylor(%,g,0,p));
expand(trigexpand(trigexpand(
expand(demoivre(%)))));
sum(coeff(%,cos(fv),k)*cf[k],k,0,p);
sum(sum(coeff(coeff(expand(%),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p);
sum(coeff(expand(%),sin(fv),k)*sf[k],k,0,p);
sum(sum(coeff(coeff(expand(%),e,l),g,k)

*eˆl*gˆk,l,0,p-k),k,0,p);
V:expand(trigreduce(expand(%)));
Lan:V-fv+f+Q;

19. Computing the Jupiter’s longitude.

Ll:fl+wl+Ql;

20. Computing the quantities v and theirs multiplications with u and ul.

v[0]:1;
v1[1]:expand(Lan-Ll-M-w-Q+Ml+wl+Ql);
for k:2 thru p do
v1[k]:sum(sum(sum(coeff(coeff(coeff(expand(
v1[k-1]*v1[1]),g,x),e,y),el,z)

*gˆx*eˆy*elˆz,x,0,p-z-y),y,0,p-z),z,0,p);

for k:1 thru p do v[k]:expand(trigreduce(v1[k]));
for k:0 thru p do for l:0 thru (p-k) do
for m:0 thru p-k-l do
uv2[k,l,m]:expand(us[k,l]*v[m]);
for k:0 thru p do for l:0 thru (p-k) do
for m:0 thru p-k-l do
uv1[k,l,m]:sum(sum(sum(coeff(coeff(coeff(
uv2[k,l,m],g,x),e,y),el,z)

*gˆx*eˆy*elˆz,x,0,p-z-y),y,0,p-z),z,0,p);
for k:0 thru p do for l:0 thru (p-k) do
for m:0 thru p-k-l do
uv[k,l,m]:expand(trigreduce(uv1[k,l,m]));
save(uv,uv);
for n:0 thru p/2 do for k:0 thru p-2*n do
for l:0 thru p-2*n-k do for m:0 thru p-2*n-k-l do
vu[n,k,l,m]:sum(sum(sum(coeff(coeff(coeff(
uv[k,l,m],g,x),e,y),el,z)*gˆx*eˆy*elˆz,
x,0,p-2*n-z-y),y,0,p-2*n-z),z,0,p-2*n);
save(vu,vu);

21. Determination the upper limit in the Fourier series.

q:2*p;

22. Calculating the S-numbers.

for m:0 thru p do for n:0 thru p do S[m,n,n]:1;
for m:0 thru p do for n:1 thru p do
S[m,n,0]:S[m,n-1,0]*(m+n);
for m:0 thru p do for n:1 thru p do
for l:1 thru n-1 do
S[m,n,l]:S[m,n-1,l]*(m+n+l)+S[m,n-1,l-1];

23. Computing the additional part in the perturbation function.

Tfi:-af*cos(fi);
for m:0 thru p do for n:0 thru p-m do
DTfi[m,n]:(-1)ˆn*sum(S[m,n,l]*afˆ(l+m)

*diff(Tfi,af,l+m),l,0,n);
Rfi:expand(trigreduce(expand(sum(sum(sum(
1/((k1-k2)!*(k2-k3)!*k3!)

*diff(DTfi[k1-k2,k2-k3],fi,k3)*
uv[k1-k2,k2-k3,k3],k3,0,k2),k2,0,k1),k1,0,p))));

24. Laplace’s coefficients depend from af.

depends(B,af);

25. Recurrence for Laplace’s coefficients.

for k:0 thru p/2-1 do for n:0 thru q do
B[2*k+3,n]:expand((afˆ2*diff(B[2*k+1,n],af,2)+
af*diff(B[2*k+1,n],af)-nˆ2*B[2*k+1,n])/
(2*k+1)ˆ2/afˆ2);

26. Fourier series.

for k:0 thru p/2 do
T[k]:(1/2*B[2*k+1,0]+
sum(B[2*k+1,n]*cos(n*fi),n,1,q));

27. Taylor series.

for k:0 thru p/2 do for m:0 thru p-2*k do
for n:0 thru p-2*k-m do
DT[k,m,n]:(-1)ˆn*sum(S[m+2*k,n,l]

*afˆ(l+m)*diff(T[k],af,l+m),l,0,n);
for k:0 thru p/2 do RA[k]:
expand(sum(sum(sum(1/((k1-k2)!*(k2-k3)!*k3!)*
diff(DT[k,k1-k2,k2-k3],fi,k3)*
vu[k,k1-k2,k2-k3,k3],
k3,0,k2),k2,0,k1),k1,0,p-2*k));
for k:0 thru p/2 do
RB[k]:expand(trigreduce(RA[k]));

28. MacLoren series.

expand(taylor((1+xˆ2)ˆ(-1/2),x,0,p));
for k:1 thru p/2 do
RC[k]:expand(coeff(%,x,2*k)

*zq[k]*(af)ˆ(2*k)*RB[k]);

29. Final computing.

for k:1 thru p/2 do for l:0 thru p do
for m:0 thru p-l do for n:0 thru p-l-m do
RD[k,n,m,l]:coeff(coeff(coeff(
RC[k],e,n),el,m),g,l);

for k:1 thru p/2 do for l:0 thru p do
for m:0 thru p-l do for n:0 thru p-l-m do
RE[k,n,m,l]:expand(trigreduce(RD[k,n,m,l]));
R:expand(RB[0]+Rfi+sum(sum(sum(sum(
RE[k,n,m,l]*eˆn*elˆm*gˆl,
n,0,p-l-m),m,0,p-l),l,0,p),k,1,p/2));
save(R,R);

4 Conclusion

This application may be easily modified for other problems. For this it must be
change only the values of p and q. In our next work, using presented application
and the generalized binomial transform [7], we will give the result of the per-
turbation function in the restricted three-body problem Sun, Jupiter and Hecuba,
containing all terms in Fourier series as far as the fourth order.

